>>>
+>+>+>+>+> +>+>+>+>+> +>+>+>+>+> +>+>+>
>>>
,[,----------]

#region Multi_ReadNumber 
+>+>+>+>+> +>+>+>+>+> +> //桁数
>+[-<
  ,+[-----------[----------------------
    [--------------<-[+<-]>[[-<+>]>]+<]
  ]]>
]
<<-[+<-]+>[->]<
#endregion
>>>>

#region Multi_ReadNumber 
+>+>+>+>+> +>+>+>+>+> +> //桁数
>+[-<
  ,+[-----------[----------------------
    [--------------<-[+<-]>[[-<+>]>]+<]
  ]]>
]
<<-[+<-]+>[->]<
#endregion
#region Multi_IsTrue
[-[+[>]>]+<] <[[<]<]>>[>]<
#endregion
>>
[
  -
  <<
  
  #region Multi_Comparate 0
  //start  ___{ a }___{ b*}___
  //result ___{ a }_x_{ b*}___
  //notice 桁数が同じである必要があります
  //x : 0 if same / 1 if a lt b / 2 if a gt b
  #region メイン比較部
  [[->>+<<]<]
  <<<[<]>
  [
    -<<+>>
    [-[-[-[-[-[-[-[-[-[
      
    ]<<[+++++++++ >>>[>]>9--------- <]>>
    ]<<[ ++++++++ >>>[>]>8--------  <]>>
    ]<<[  +++++++ >>>[>]>7-------   <]>>
    ]<<[   ++++++ >>>[>]>6------    <]>>
    ]<<[    +++++ >>>[>]>5-----     <]>>
    ]<<[     ++++ >>>[>]>4----      <]>>
    ]<<[      +++ >>>[>]>3---       <]>>
    ]<<[       ++ >>>[>]>2--        <]>>
    ]<<[        + >>>[>]>1-         <]>>
    ]<<[          >>>[>]>0          <]>>
    
    >[>]>>
    
    -<<+>>
    [-[-[-[-[-[-[-[-[-[
      
    ]<<[+++++++++ [<]<9+++++++++ <]>>
    ]<<[ ++++++++ [<]<8++++++++  <]>>
    ]<<[  +++++++ [<]<7+++++++   <]>>
    ]<<[   ++++++ [<]<6++++++    <]>>
    ]<<[    +++++ [<]<5+++++     <]>>
    ]<<[     ++++ [<]<4++++      <]>>
    ]<<[      +++ [<]<3+++       <]>>
    ]<<[       ++ [<]<2++        <]>>
    ]<<[        + [<]<1+         <]>>
    ]<<[          [<]<0          <]>>
    
    
    
    <
    [
      <->>+<
      -[-[-[-[-[-[-[-[-[-[[+]>+<]]]]]]]]]]
    ]
    <+
    [- <[<]]>
  ]
  >[<]>><<
  >[-<+>]<
  #endregion
  #region 後処理部
  <<[<]<<[[->>+<<]<]>>>[>]>>>[>]>>[[-<<+>>]>]<<<
  #endregion
  #endregion
  
  [<]+<
  
  [-[-[-
    
  ]>[-
    
    <<<
    #region Multi_AddToPrev 0
    //start  ___{ a }___{ b*}___
    //result ___{res}___{ b*}___
    [
      [<]<[<]<<[->>+<<]>>[>]>[>]>+<<-
      [ [<]<[<]>+[>]>[>] >+<<- ] <
    ]
    <[<]<<[<]>
    [[-<+>]>]>>[[-<<<+>>>]>]<<<<
    [ 
      [->+<[->+<[->+<[->+<[->+<[->+<[->+<[->+<[->+<[->+<[ 
        [->+<]>----------< <[+>]>[<]
      ]]]]]]]]]]]<
    ]
    >>[>]>>>>>[[-<<+>>]>]<<<
    #endregion
    
    >>>>[>]<
    
    #region Multi_SubFromPrev 1
    //start  ___{ a }___{ b*}___
    //result ___{res}___{ b*}___
    //notice a の桁が b 以上である必要があります。
    //       アンダーフロー処理済み
    [<]{<<<[<]}<<<[<]++[[-<+>]>]
    >>>{[>]>>>}[>]<
    [
      ->>+<<
      [ [<]{<<<[<]}<<<[<]<[-[>[+++++++++>]>[>]>]+<] >-< >>>{[>]>>>}[>]>+<<- ]
      <[<]{<<<[<]}<<<[<]<[->+<]>[>]>>>{[>]>>>}[>]<
    ]
    {<<<[<]}<<<[<]<[[->+<]<]>>[-]>[>]>>>{[>]>>>}>>[[-]<<+>>>]<<<
    #endregion
    >>
  ]<
  ]>[- 
    <<<[<]>[[-]>]
    
    >>>[ [-<<<<<<<<<<<<<<+>>>>>>>>>>>>>>]+> ]
    >
  ]<
  ]>[- 
    
    >[[-]+>]
    >
  ]<
  
  @
  
  
  <
    [[-]<]>
  
  #region Multi_ReadNum
  <
  +>+>+>+>+>+>+>+>+>+>+[<]>><
  [-<
    ,+[-----------[----------------------
      [--------------->[-]]
    ]]>
  ]
  
  <<[
    >>+[>]<-<[<]<
    [->>[>]>+<<[<]<]
    <
  ]>>>[>]+[>]<
  #endregion
  
  
  #region Multi_IsTrue
  [-[+[>]>]+<] <[[<]<]>>[>]<
  #endregion
  >>
]
<<[<]<<<[<]<<<

#region Multi_WriteNumber
+[<]>[-[+[>]]<+>>] ++++++[-<-------->]<+<--> [[<]>[+>]<]<[<]>[.>]
>++++++[-<++++++++>]<- [[<]>[->]<]<[<]<[[->+<]<]>>[>]<　//数値復帰部
#endregion
